

# 最大エントロピー法と AIC


溝口卓哉\*・美谷島 実\*\*

## 1 はじめに

クォーク・グルーオン・プラズマ (Quark Gluon Plasma; QGP) のシグナルの研究は、高エネルギー重イオン反応物理学における大切な課題の一つである。そのため、私たちは高多重度分布として JACEE events を用いて、これらの中に QGP の信号が入っているか否かのテストに、最大エントロピー法 (Maximum Entropy Method; MEM) [1] を使うことを考えた [2, 3]。MEM 法の次数を決めるために、AIC (Akaike Information Criterion) が有用であることが知られている [1]。ところが MEM および AIC は、高エネルギー物理や重イオン物理学者になじみが薄いので、ここで「スペクトル解析」(日野著) [1] に従って、JACEE データを用いて、一步一步計算をたどってみる。MEM, AIC, パワースペクトル (Power Spectra: PS) 解析等の間の関係を調べたい。用いられた方程式に (H ○○) とあるのは文献 [1] 中の式の番号を表わす。

## 2 JACEE events

1983年 JACEE events [4] が発表された。これらは高多重度分布のデータで、図 1 に示されている。横軸は pseudorapidity( $\eta$ ), 縦軸は分布数  $dN/d\eta$  である。反応は Ca-C, Fe-Pb, Si-AgBr である。



\* Takuya MIZOGUCHI (Toba National College of Maritime Technology, Toba 517)

\*\* Minoru BIYAJIMA (Department of Physics, Faculty of Liberal Arts, Shinshu University, Matsumoto 390)




図1 JACEE Events, a) Ca-C, b) Fe-Pb, 及び c) Si-AgBr 衝突.

### 3 MEM 法

MEM 法の計算機プログラムリストを付録に示す。これは文献 [1], 235~236項のプログラムを N88-BASIC 用書き換えたものである。

## 付録 MEM法のプログラムリスト

```

1000  DEFINIT I-N
1010  DEFDBL B,C,F,G,S
1020  'PARAMETER
1030    NMAX=2001 : MMAX=100 : LMAX=1001 : NYQ=1000
1040  '
1050  '*****
1060  'MEM ( MAX ENTROPY METHOD )
1070  '*****
1080  '
1090  DIM B1(NMAX),B2(NMAX),C(LMAX),F(NYQ),G(MMAX),GG(MMAX)
1100  DIM X(NMAX),E(NYQ),FPE(MMAX),AIC(MMAX)
1110  '
1120  GOSUB *READ1
1130  '
1140  GOSUB *BURG
1150  '
1160  GOSUB *MEM
1170  '
1180  GOSUB *OUTPUT
1190  '
1200  STOP
1210  END
1220  '*****
1230  *READ1
1240  '
1250  'X(I) = INPUT DATA, (I=1,NMAX)
1260  'G(I) = PREDICTION ERROR COEFFICIENT
1270  'C(I) = A.C. COEFFS. AT T=(I-1)*DT, (I=1,LMAX)
1280  'FPE = FINAL PREDICTION ERRORS
1290  'AIC = AKAIKE'S INFORMATION CRITERION
1300  'PM = OUTPUT POWER FROM THE PREDICTION ERROR FILTER
1310  '*****
1320  '
1330  INPUT "DATA FILE : ",DATAFILES$
1340  INPUT "NMAX : ",NMAX
1350  INPUT "MMAX : ",MMAX
1360  INPUT "LMAX : ",LMAX
1370  INPUT "DT : ",DT
1380  '
1390  IF MMAX < NMAX-1 GOTO *L10
1400  PRINT "MMAX SHOULD BE SMALLER THAN NMAX-1" : STOP
1410  *L10
1420  NYQ=INT((NMAX-1)/2)
1430  '
1440  OPEN DATAFILES$ FOR INPUT AS #1
1450  FOR I=1 TO NMAX : INPUT #1, X(I) : NEXT I
1460  CLOSE
1470  '
1480  SUMX=0!
1490  FOR I=1 TO NMAX : SUMX=SUMX+X(I) : NEXT I
1500  AVX=SUMX/NMAX
1510  FOR I=1 TO NMAX : X(I)=X(I)-AVX : NEXT I
1520  '
1530  SUM=0!
1540  FOR I=1 TO NMAX : SUM=SUM+X(I)^2 : NEXT I
1550  C(1)=SUM/NMAX
1560  PM=C(1)
1570  '
1580  B1(1)=X(1)
1590  FOR I=2 TO NMAX
1600  B1(I)=X(I)
1610  B2(I-1)=X(I)
1620  NEXT I
1630  '
1640  RETURN
1650  '*****
1660  *BURG
1670  '
1680  'COMPUTATION OF G'S (PREDICTION ERROR COEFFS.) BY THE LEVINSON
1690  ' ALGORITHM AND FPE & AIC
1700  '

```

```

1710 FOR M=1 TO MMAX
1720 '
1730 STN=0!
1740 STD=0!
1750 FOR I=1 TO NMAX-M
1760 STN=STN+B1(I)*B2(I)
1770 STD=STD+B1(I)^2+B2(I)^2
1780 NEXT I
1790 G(M)=-2!*STN/STD
1800 PM=PM*(1!-G(M)^2)
1810 '
1820 IF M=1 GOTO *L20
1830 FOR K=1 TO M-1 : G(K)=GG(K)+G(M)*GG(M-K) : NEXT K
1840 *L20
1850 FOR I=1 TO NMAX-M-1
1860 B1(I)=B1(I)+G(M)*B2(I)
1870 B2(I)=B2(I+1)+G(M)*B1(I+1)
1880 NEXT I
1890 '
1900 FOR I=1 TO M : GG(I)=G(I) : NEXT I
1910 '
1920 ' -----
1930 '( ESTIMATION OF A.C. COEFF.)
1940 SUM=0!
1950 FOR I=1 TO M : SUM=SUM-C(M+1-I)*G(I) : NEXT I
1960 C(M+1)=SUM
1970 '
1980 ' -----
1990 '( FPE AND AIC )
2000 IF M = NMAX-1 GOTO *L30
2010 FPE(M)=(NMAX+M+1)/(NMAX-M-1)*PM
2020 AIC(M)=NMAX*LOG(2!*3.14159*PM)+NMAX+2*(M+1)
2030 *L30
2040 NEXT M
2050 '
2060 RETURN
2070 '*****
2080 *MEM
2090 '
2100 FO=.1
2110 FOR I=1 TO NYQ
2120 F(I)=FO*I
2130 SUMR=1
2140 SUMI=0
2150 CI=2*3.14159*F(I)*DT
2160 FOR J=1 TO MMAX
2170 SUMR=SUMR+G(J)*COS(CI*J)
2180 SUMI=SUMI+G(J)*SIN(CI*J)
2190 NEXT J
2200 '
2210 E(I)=2*PM*DT/(SUMR^2+SUMI^2)
2220 NEXT I
2230 '
2240 ' -----
2250 '( EXTRAPOLATION OF A.C. BY MEM )
2260 IF MMAX >= LMAX GOTO *L40
2270 FOR L=MMAX+1 TO LMAX
2280 SUM=0!
2290 FOR I=1 TO MMAX : SUM=SUM-C(L+1-I)*G(I) : NEXT I
2300 C(L+1)=SUM
2310 NEXT L
2320 *L40
2330 RETURN
2340 '*****
2350 *OUTPUT
2360 '
2370 PRINT " I "; " FRQ "; " SPC "; " A.C. "; " FPE
"; "
"; " AIC"
2380 FOR I=1 TO MMAX
2390 PRINT USING "###";I;
2400 PRINT " "; :PRINT USING "##.###^";F(I);
2410 PRINT " "; :PRINT USING "##.###^";E(I);
2420 PRINT " "; :PRINT USING "##.###^";C(I);

```

```

2430 PRINT " "; :PRINT USING "##.###^####";FPE(I);
2440 PRINT " "; :PRINT USING "##.###^####";AIC(I)
2450 NEXT I
2460 IF NYQ >= LMAX THEN I1=LMAX
2470 IF NYQ < LMAX THEN I1=NYQ
2480 IF MMAX = I1 GOTO *L50
2490 FOR I=MMAX+1 TO I1
2500 PRINT USING "###";I;
2510 PRINT " "; :PRINT USING "##.###^####";F(I);
2520 PRINT " "; :PRINT USING "##.###^####";E(I);
2530 PRINT " "; :PRINT USING "##.###^####";C(I)
2540 NEXT I
2550 *L50
2560 IF NYQ = LMAX THEN RETURN
2570 IF NYQ > LMAX GOTO *L60
2580 FOR I=NYQ+1 TO LMAX
2590 PRINT USING "###";I;
2600 PRINT " ";
2610 PRINT USING "##.###^####";C(I)
2620 NEXT I
2630 *L60
2640 FOR I=LMAX+1 TO NYQ
2650 PRINT USING "###";I;
2660 PRINT " "; :PRINT USING "##.###^####";F(I);
2670 PRINT " "; :PRINT USING "##.###^####";E(I)
2680 NEXT I
2690 '
2700 RETURN
    
```

#### 4 「スペクトル解析」[1] との対応

Si-AgBr を用いて MEM スペクトル( $m=3$ )を求める。ただし、AIC は次の式を用いる [3]。

$$(AIC)_m = N \ln(2\pi P_m) + N + 2(m + 1), \quad (m = 1, 2, \dots) \quad 4-1)$$

下記の JACEE event を用いて順次計算を実行する。X(i)を時系列として取り扱っている。

表1 JACEE event Si-AgBr の時系列

DATA x(1)~x(96)										
i	1	2	3	4	5	6	7	8	9	10
x(i)	2.0	4.0	4.0	6.0	4.0	8.0	7.0	5.0	10.0	5.0
i	11	12	13	14	15	16	17	18	19	20
x(i)	7.0	1.0	7.0	11.0	14.0	8.0	13.0	11.0	10.0	15.0
i	21	22	23	24	25	26	27	28	29	30
x(i)	10.0	12.0	15.0	20.0	19.0	9.0	12.0	14.0	14.0	13.0
i	31	32	33	34	35	36	37	38	39	40
x(i)	19.0	16.0	17.0	12.0	7.0	17.0	22.0	22.0	24.0	16.0
i	41	42	43	44	45	46	47	48	49	50
x(i)	20.0	16.0	21.0	16.0	12.0	17.0	18.0	11.0	21.0	23.0
i	51	52	53	54	55	56	57	58	59	60
x(i)	15.5	30.5	28.0	19.5	16.5	14.0	16.0	15.5	13.5	14.0
i	61	62	63	64	65	66	67	68	69	70
x(i)	7.0	12.5	14.5	12.5	12.5	8.5	8.5	8.0	12.5	11.0
i	71	72	73	74	75	76	77	78	79	80
x(i)	8.5	10.0	10.5	15.0	8.5	7.5	8.0	6.7	7.0	4.2
i	81	82	83	84	85	86	87	88	89	90
x(i)	4.4	2.5	2.2	1.3	2.2	1.8	0.7	0.5	0.3	0.3
i	91	92	93	94	95	96				
x(i)	1.2	0.0	0.0	0.0	0.0	1.2				

$$\begin{aligned}
\mu &= \sum_{i=1}^{96} x(i)/96 \\
&= 1.075E + 01 \\
C(0) = P0 &= \sum_{i=1}^{96} [x(i) - \mu]^2/96 \\
&= 4.691D + 01
\end{aligned} \tag{H12.98}$$

$$\begin{aligned}
b(1, i) &= x(i) - \mu \\
b'(1, i) &= x(i+1) - \mu
\end{aligned} \tag{H12.100a}$$

$$\begin{aligned}
\gamma(1, 1) &= -2 \sum_{i=1}^{96-1} b(1, i) \cdot b'(1, i) / \sum_{i=1}^{96-1} [b^2(1, i) + b'^2(1, i)] \\
&= -8,166D - 01
\end{aligned} \tag{H12.99}$$

$$\begin{aligned}
P1 &= P0 \times [1 - \gamma^2(1, 1)] \\
&= 1.563E + 01
\end{aligned} \tag{H12.103}$$

$$\begin{aligned}
C(1) &= -\gamma(1, 1) \cdot C(0) \\
&= 3.831D + 01
\end{aligned} \tag{H12.104}$$

$$\begin{aligned}
FPE(1) &= \frac{96 + (1+1)}{96 - (1+1)} \times P1 \\
&= 1.629D + 01
\end{aligned} \tag{H12.105}$$

$$\begin{aligned}
AIC(1) &= 96 \times \ln(2\pi \times P1) + 96 + 2 \times (1+1) \\
&= 5.404E + 02
\end{aligned} \tag{4-(1)}$$

$$\begin{aligned}
b(2, i) &= b(1, i) + \gamma(1, 1) \cdot b'(1, i) \\
b'(2, i) &= b'(1, i+1) + \gamma(1, 1) \cdot b(1, i+1)
\end{aligned} \tag{H12.100}$$

$$\begin{aligned}
\gamma(2, 2) &= -2 \sum_{i=1}^{96-2} b(2, i) \cdot b'(2, i) / \sum_{i=1}^{96-2} [b^2(2, i) + b'^2(2, i)] \\
&= -2.501D - 01
\end{aligned} \tag{H12.99}$$

$$\begin{aligned}
P2 &= P1 \times [1 - \gamma^2(2, 2)] \\
&= 1.465E + 01
\end{aligned} \tag{H12.103}$$

$$\begin{aligned}
\gamma(2, 1) &= \gamma(1, 1) + \gamma(2, 2) \cdot \gamma(1, 1) \\
&= -6.124D - 01
\end{aligned} \tag{H12.102}$$

$$\begin{aligned}
C(2) &= -[\gamma(2, 1) \cdot C(1) + \gamma(2, 2) \cdot C(0)] \\
&= 3.519D + 01
\end{aligned} \tag{H12.104}$$

$$\begin{aligned}
FPE(2) &= \frac{96 + (2+1)}{96 - (2+1)} \times P2 \\
&= 1.560D + 01
\end{aligned} \tag{H12.105}$$

$$\begin{aligned}
AIC(2) &= 96 \times \ln(2\pi \times P1) + 96 + 2 \times (2+1) \\
&= 5.362E + 02
\end{aligned} \tag{4-(1)}$$

$$\begin{aligned}
b(3, i) &= b(2, i) + \gamma(2, 2) \cdot b'(2, i) \\
b'(3, i) &= b'(2, i+1) + \gamma(2, 2) \cdot b(2, i+1)
\end{aligned} \tag{H12.100}$$

$$\begin{aligned}
\gamma(3, 3) &= -2 \sum_{i=1}^{96-3} b(3, i) \cdot b'(3, i) / \sum_{i=1}^{96-3} [b^2(3, i) + b'^2(3, i)] \\
&= -2.210D - 01
\end{aligned} \tag{H12.99}$$

$$\begin{aligned}
P3 &= P2 \times [1 - \gamma^2(3, 3)] \\
&= 1.394E + 01
\end{aligned} \tag{H12.103}$$

$$\gamma(3, 1) = \gamma(2, 1) + \gamma(3, 3) \cdot \gamma(2, 2) \tag{H12.102}$$

$$\begin{aligned}
 &= -5.571D - 01 \\
 \gamma(3, 2) &= \gamma(2, 2) + \gamma(3, 3) \cdot \gamma(2, 1) && \text{(H12.102)} \\
 &= -1.147D - 01 \\
 C(3) &= -[\gamma(3, 1) \cdot C(2) + \gamma(3, 2) \cdot C(1) + \gamma(3, 3) \cdot C(0)] && \text{(H12.104)} \\
 &= 3.437D + 01 \\
 \text{FPE}(3) &= \frac{96 + (3 + 1)}{96 - (3 + 1)} \times P3 && \text{(H12.105)} \\
 &= 1.515D + 01 \\
 \text{AIC}(3) &= 96 \times \ln(2\pi \times P1) + 96 + 2 \times (3 + 1) && \text{(4-(1))} \\
 &= 5.333E + 02
 \end{aligned}$$

$$P(f) = \frac{0.1 \times P3}{\left| 1 + \sum_{k=1}^3 \gamma(3, k) \cdot \exp(i2\pi f \cdot k \times 0.1) \right|^2} \quad \text{(H12.107)}$$

f	1.000D-01	2.000D-01	3.000D-01	4.000D-01	5.000D-01
P(f)	1.326E+02	5.645E+01	2.913E+01	1.757E+01	1.180E+01
f	6.000D-01	7.000D-01	8.000D-01	9.000D-01	1.000D+00
P(f)	8.537E+00	6.534E+00	5.222E+00	4.321E+00	3.681E+00
f	1.100D+00	1.200D+00	1.300D+00	1.400D+00	1.500D+00
P(f)	3.212E+00	2.864E+00	2.601E+00	2.402E+00	2.252E+00
f	1.600D+00	1.700D+00	1.800D+00	1.900D+00	2.000D+00
P(f)	2.140E+00	2.059E+00	2.003E+00	1.969E+00	1.953E+00
f	2.100D+00	2.200D+00	2.300D+00	2.400D+00	2.500D+00
P(f)	1.952E+00	1.965E+00	1.989E+00	2.020E+00	2.056E+00
f	2.600D+00	2.700D+00	2.800D+00	2.900D+00	3.000D+00
P(f)	2.093E+00	2.127E+00	2.152E+00	2.164E+00	2.160E+00
f	3.100D+00	3.200D+00	3.300D+00	3.400D+00	3.500D+00
P(f)	2.137E+00	2.093E+00	2.030E+00	1.952E+00	1.862E+00
f	3.600D+00	3.700D+00	3.800D+00	3.900D+00	4.000D+00
P(f)	1.766E+00	1.668E+00	1.571E+00	1.480E+00	1.395E+00
f	4.100D+00	4.200D+00	4.300D+00	4.400D+00	4.500D+00
P(f)	1.318E+00	1.250E+00	1.191E+00	1.141E+00	1.099E+00
f	4.600D+00	4.700D+00			
P(f)	1.066E+00	1.040E+00			

SIMULTANEOUS EQUATIONS (H12.84)

$$C(0) = P0$$

$$4.691D+01 = 4.691E+01$$

$$\begin{vmatrix} C(0) & C(1) \\ C(1) & C(0) \end{vmatrix} \begin{vmatrix} 1 \\ \gamma(1,1) \end{vmatrix} = \begin{vmatrix} P1 \\ 0 \end{vmatrix}$$

$$\begin{vmatrix} 4.691D+01 & 3.831D+01 \\ 3.831D+01 & 4.691D+01 \end{vmatrix} \begin{vmatrix} 1 \\ -8.166D-01 \end{vmatrix} = \begin{vmatrix} 1.563E+01 \\ 0 \end{vmatrix}$$

$$\begin{vmatrix} C(0) & C(1) & C(2) \\ C(1) & C(0) & C(1) \\ C(2) & C(1) & C(0) \end{vmatrix} \begin{vmatrix} 1 \\ \gamma(2,1) \\ \gamma(2,2) \end{vmatrix} = \begin{vmatrix} P2 \\ 0 \\ 0 \end{vmatrix}$$

$$\begin{vmatrix} 4.691D+01 & 3.831D+01 & 3.519D+01 \\ 3.831D+01 & 4.691D+01 & 3.831D+01 \\ 3.519D+01 & 3.831D+01 & 4.691D+01 \end{vmatrix} \begin{vmatrix} 1 \\ -6.124D-01 \\ -8.166D-01 \end{vmatrix} = \begin{vmatrix} 1.465E+01 \\ 0 \\ 0 \end{vmatrix}$$

$$\begin{array}{c}
 \left| \begin{array}{cccc|c|c}
 C(0) & C(1) & C(2) & C(3) & 1 & P3 \\
 C(1) & C(0) & C(1) & C(2) & \gamma(3,1) & 0 \\
 C(2) & C(1) & C(0) & C(1) & \gamma(3,2) & 0 \\
 C(3) & C(2) & C(2) & C(0) & \gamma(3,2) & 0
 \end{array} \right| \\
 \left| \begin{array}{cccc|c|c}
 4.691D+01 & 3.831D+01 & 3.519D+01 & 3.437D+01 & 1 & 1.394E+01 \\
 3.831D+01 & 4.691D+01 & 3.831D+01 & 3.519D+01 & -6.124D-01 & 0 \\
 3.519D+01 & 3.831D+01 & 4.691D+01 & 3.831D+01 & -8.166D-01 & 0 \\
 3.437D+01 & 3.519D+01 & 3.831D+01 & 4.691D+01 & -8.166D-01 & 0
 \end{array} \right|
 \end{array}$$

$$C(k) = -\gamma(3, 1) \cdot C(k-1) - \gamma(3, 2) \cdot C(k-2) - \gamma(3, 3) \cdot C(k-3) \quad (k \geq 4)$$

(H12.108)

k	0	1	2	3	4
C(k)	4.691D+01	3.831D+01	3.519D+01	3.437D+01	3.166D+01
k	5	6	7	8	9
C(k)	2.936D+01	2.759D+01	2.573D+01	2.399D+01	2.242D+01
k	10	11	12	13	14
C(k)	2.093D+01	1.953D+01	1.824D+01	1.703D+01	1.590D+01
k	15	16	17	18	19
C(k)	1.484D+01	1.386D+01	1.294D+01	1.208D+01	1.127D+01
k	20	21	22	23	24
C(k)	1.053D+01	9.827D+00	9.175D+00	8.566D+00	7.997D+00
k	25	26	27	28	29
C(k)	7.466D+00	6.970D+00	6.507D+00	6.075D+00	5.672D+00
k	30	31	32	33	34
C(k)	5.295D+00	4.944D+00	4.615D+00	4.309D+00	4.023D+00
k	35	36	37	38	39
C(k)	3.756D+00	3.506D+00	3.273D+00	3.056D+00	2.853D+00
k	40	41	42	43	44
C(k)	2.664D+00	2.487D+00	2.322D+00	2.168D+00	2.024D+00
k	45	46	47	48	49
C(k)	1.889D+00	1.764D+00	1.647D+00	1.537D+00	1.435D+00
k	50				
C(k)	1.340D+00				

### 5 AIC の次数と PS の関係

図 2 に JACEE events の AIC を示す。図 3 に Si-AgBr の PS と次数の関係を示す。AIC の要求するところは、AIC が最小の PS が意味あるということで、Fe-Pb: m=14, Si-AgBr: m=5, Ca-C: m=7 が決定され、それらの PS を図 4 に示す。フラクタル次元 D も示されている [2]。次に m を増加させた時の PS を図 5 に示す。

表 2 に P(f) の極大と f の値の関係が示されている。Takagi [5] の PS も参考のために示されている。






図2 JACEE Events の AIC.











図3-a)



☒ 3-b)










図 3-e)

図 3 Si-AgBr の PS と次数の関係。次数  $m=1 \sim 10$ 。









図 4 AIC の極小点と PS  $P(f)$ . a) Ca-C, 及び b) Fe-Pb.



☒ 5-a)






図 5 - b)





図 5 - c)

図 5 高次数  $m$  の値と PS. a) Ca-C, b) Fe-Pb, 及び c) Si-AgBr 衝突.

表2 P(f)の極大と f の値の関係。ω[Takagi]=f.

Fe-Pb m=20		Fe-Pb m=30			
f	P(f)	f	P(f)		
0.14	3616.11	0.14	3552.2		
0.71	2.91911	0.77	4.04967		
1.22	2.96771	1.22	2.99024		
1.73	12.862	1.77	15.3936		
2.41	9.45512	2.33	11.006		
3.23	3.4756	2.56	9.04471		
4.11	8.79189	3.22	5.4817		
4.79	4.27388	4.06	12.2234		
		4.72	4.66318		
Si-AgBr m=20		Si-AgBr =30		Si-AgBr(Takagi)	
f	P(f)	f	P(f)	ω	φ(ω)
0.11	342.701	0.11	1069.09	0.244	429
0.74	10.5505	0.53	9.91823	0.588	471
1.36	5.20732	0.84	9.49953	0.850	431
2.02	4.81546	1.40	8.84824		
3.09	4.28979	1.89	5.75317		
3.73	6.57818	2.85	3.81123		
4.35	2.73194	3.16	6.94659		
		3.72	7.91237		
		4.29	2.46357		
Ca-C m=20		Ca-C m=30		Ca-C(Takagi)	
f	P(f)	f	P(f)	ω	φ(ω)
0.01	65.2552	0.10	95.1593	0.125	324
1.41	2.93465	0.53	4.46775	1.17	254
2.33	2.47222	1.18	3.34665	4.11	257
3.00	1.68936	1.50	5.04392		
3.63	1.87593	2.27	1.88294		
4.17	3.32691	3.00	2.334		
		3.65	2.49044		
		4.11	4.53469		

## 6 結 論



MEM法の次数の決定にAICが有効というのは図3から分かった。MEM法を使うとPSの雑音の部分が消えるのは図4から分かった。Takagiの解析とSi-AgBrでm=30程度が一致することを見出したがm=30の意味が不明であるので考察したい。



## 参考文献

- [1] 日野幹雄, スペクトル解析 (朝倉書店, 1979).  
 [2] N. Suzuki, Y. Yasutani, M. Biyajima, and T. Mizoguchi, Prog. Theor. Phys. 85 (1991), 149.  
 [3] N. Suzuki and M. Biyajima, Prog. Theor. Phys. 88 (1992), 73.  
 [4] T. H. Burnett et al., JACEE Collaboration, Phys. Rev. Lett., 50 (1983), 2062.  
 [5] F. Takagi, Phys. Rev. Lett. 53 (1984), 427.

[追記] 著者達は, FFT (Fast Fourier Transform) と MEM 法との比較から,  $m$  を決める方法があることを知っていた。ここで Takagi [5] の図と我々の FFT の結果を示す。 $m=30$ 程度にとると MEM 法, PS, FFT でのピーク値がほぼ一致することがこれらの比較から分かってきた。



The solid curve is the power spectrum  $\phi(\omega)$  of the Si-AgBr event observed by JACEE. The dashed and the dash-dotted curves are, respectively, the mean power spectrum  $\langle\phi(\omega)\rangle$  and  $\langle\phi(\omega)\rangle + D_\phi(\omega)$  obtained from the simulation. Takagi [5].

